CIS 120 Midterm I February 15, 2013

Name (printed): _____
Pennkey (login id): _____

My signature below certifies that I have complied with the University of Pennsylvania's Code of Academic Integrity in completing this examination.

Signature: _____ Date: _____

SOLUTIONS

1. Program Design (27 points total)

Use the four-step design methodology to implement a function called `insert` that takes an `int` and a list of ints and inserts the number into the list at the first position where it is less than or equal to the next number. If the number is greater than all others in the list, it should be added to the end. If the list is sorted before the call, the result will also be sorted.

For example, `insert 3 [1; 2; 4; 5]` should yield the list `[1; 2; 3; 4; 5]`.

(0 points) Step 1 is *understanding the problem*. You don't have to write anything for this part—your answers below will demonstrate whether or not you succeeded with Step 1.

(3 points) Step 2 is *formalizing the interface*. Write down the *type* of the `insert` function as you might find it in a `.mli` file or module interface.

```
val insert: int -> int list -> int list
```

Grading Scheme.

- *Generic type* `'a -> 'a list -> 'a` *list also ok*
- *-1 wrong number of arguments*
- *-1 wrong type of arguments*
- *-1 doesn't return a list*

(12 points) Step 3 is *writing test cases*. Complete the following three tests with the expected behavior. We have done the first one for you, based on the problem description.

Note that some test cases are better than others, and credit will be assigned accordingly: make sure your tests cover a sufficiently broad range of "interesting" input numbers and lists. Fill in the description string of the `run_test` function with a short explanation of *why* the test case is interesting. Your description should not just restate the test case, e.g. "insert 3 [1;2;4;5]".

i.
```
let test () : bool =
  insert 3 [1;2;4;5] = [1;2;3;4;5]
;; run_test "insert into middle of list" test
```

Good answers:

```
(1)  insert 0 [] = [0]
```

Check the Nil case

```
(2)  insert 0 [1;2] = [0;1;2]
```

Insert at beginning

```
(3)  insert 3 [1;2] = [1;2;3]
```

Insert at end

```
(5)  insert 3 [1;2;3] = [1;2;3;3]
```

Insert duplicate

Grading Scheme. 4 points per test case.

- *-1 wrong answer to test*
- *-4 not "interesting" (duplicate)*
- *-1 poor or no description (i.e. description just states what the test case is "insert 3", not why it was interesting.)*

(12 points) Step 4 is *implementing the program*. Fill in the body of the `insert` function to complete the design. Do *not* use any list library functions (such as `fold`, or `@`) to solve this problem. If you would like to use a helper function in your answer, you must define it.

```
let rec insert (x: int) (xs : int list) : int list =
      begin match xs with
      | [] -> [ x ]
      | y :: ys -> if x <= y then x :: y :: ys else
                   y :: insert x ys
      end
```

Grading scheme:

- *no deduction for minor syntax errors*
- *-1 wrong or missing type annotations*
- *-2 incorrect Nil case*
- *-2 incorrect test comparing x and y*
- *-2 (each) omitting x or y from insertion case*
- *-2 omitted y from noninsertion case*
- *-2 Not recursing on correct list*
- *-2 assuming the list contains no duplicates*
- *-2 insert in multiple cases*
- *various other errors at discretion*

2. List recursion, higher-order functions and generic types (29 points total)

This problem considers the following function, called `separate`.

```
let rec separate (v:int) (lst : int list) : int list * int list =
 begin match lst with
 | [] -> ([],[])
 | hd :: tl ->
   let (xs,ys) = separate v tl in
   if hd >= v then
     (xs, hd :: ys)
   else
     (hd :: xs, ys)
 end
```

a. (9 points) Complete the following test cases for `separate` so that they return true.

```
let test () : bool =

 separate 5 [] = _____________([],[])_________________

let test () : bool =

 separate 5 [1;3;6;7] = ___([1;3],[6;7])_____________

let test () : bool =

 separate 5 [1;5;6] = _______([1],[5;6])____________
```

Grading Scheme: 3 points per blank. No partial credit. No deduction for minor syntax errors, i.e. commas vs. semicolons as long as the answer is unambiguous.

b. (9 points) Now consider a version of `separate`, called `ho_separate`, that takes a higher-order function as an additional argument. Here are two test cases for this version.

```
let nonnegative (x:int):bool = x >= 0
let test () : bool =
  ho_separate nonnegative [-1; 1; 0; -2] = ([-1; -2],[1;0])

let positive (x:int):bool = x > 0
let test () : bool =
 ho_separate positive [-1; 0; 2; -2] = ([-1; 0; -2],[2])
```

Fill in the blanks to complete the implementation of `ho_separate`.

```
let rec ho_separate (f : _____int -> bool_______)
                (lst : int list) : int list * int list =
 begin match lst with
 | [] -> ([],[])
 | hd :: tl ->

   let (xs,ys) = ___ho_separate f tl_____ in


   if _____f hd_______ then
     (xs, hd :: ys)
   else
     (hd :: xs, ys)
 end
```

c. (4 points) Reimplement `separate` using `ho_separate` as a helper function. You should not use recursion—just call `ho_separate` with the appropriate arguments.

```
let separate (v:int) (lst : int list) : int list * int list =
  ho_separate (fun x -> x >= v) lst
```

Grading Scheme:

- *-1 for wrong comparison operator (such as equality)*
- *-1 if missing lst argument*
- *-3 if missing function argument*
- *-2 if h.o.f. takes wrong number of arguments*
- *-1 if missing ho_separate*
- *other errors at discretion*

d. (3 points) Now consider a different version of `separate`, called `generic_separate`. Here are two test cases for `generic_separate`.

```
let test () : bool =
 generic_separate "c" ["a";"b";"d"] = (["a";"b"],["d"])

let test () : bool =
 generic_separate 0.5 [0.0;0.7;0.2; 0.8] = ([0.0;0.2],[0.7;0.8])
```

(Note that `generic_separate` does *not* take a higher-order function as an argument.)
What is the interface to this function? Write the type as it might appear in a `.mli` file.

```
val generic_separate: 'a -> 'a list -> 'a list * 'a list
```

Grading Scheme: 1 point partial credit for answer `'a -> 'a list -> 'a list`.

e. (4 points) Reimplement `separate` using `generic_separate` as a helper function. You should not use recursion—just call `generic_separate` with the appropriate arguments.

```
let separate (v:int) (lst : int list) : int list * int list =
  generic_separate v lst
```

Grading Scheme:

- *-1 if missing lst argument*
- *-1 if missing v argument*
- *-1 if missing generic_separate*
- *other errors at discrection*

3. Types (16 points)

For each OCaml value or function definition below, fill in the blank where the type annotation could go or write "ill typed" if there is a type error. If an expression can have multiple types, give the most generic one. Recall that the `@` operator appends two lists together in OCaml. We have done the first one for you. Consider the definitions to be below the following code:

```
module type SET = sig
  type 'a set
  val fromList : 'a list -> 'a set
end

module LSet : SET = struct
  type 'a set = 'a list
  let fromList (l : 'a list) = l
end

open LSet;;

let x : _______ string _______________ = "120 " ^ "is fun"


let a : ____________ill typed_____________________ = "120" ^ 120

let b : ____________ill typed_____________________ = [120] :: [120]

let c : _____________int list_____________________ = 120 :: [120]

let d : ____________int * int_____________________ = (120, 120)

let e : _________(int * int) list_________________ = [(120, 120)]

let f : __________int set_____________________ = fromList [120]

let g : __________int set_____________________ = fromList ([2] @ [3])

let h : ___________ill typed_____________________ = (fromList [2]) @ (fromList [3])
```

Grading scheme: 2 points per answer: 0 if wrong, 2 if right. No deduction for omitting parens on (e).

4. Binary Search Trees (28 points total)

Recall the definition of generic binary trees and the binary search tree `insert` function:

```
type 'a tree =
   | Empty
   | Node of 'a tree * 'a * 'a tree

let rec insert (t:'a tree) (n:'a) : 'a tree =
 begin match t with
   | Empty -> Node(Empty, n, Empty)
   | Node(lt, x, rt) ->
      if x = n then t
      else if n < x then Node (insert lt n, x, rt)
      else Node(lt, x, insert rt n)
 end
```

a. (5 points) Circle the trees that satisfy the *binary search tree invariant*. (Note that we have omitted the Empty nodes from these pictures.)



Answer: `(a), (b)`

b. (8 points) For each definition below, circle the letter of the tree above that it constructs or "none of the above".

Grading Scheme: 4 points per answer

```
let t1 : int tree =
 Node(Node(Empty, 1, Empty), 2, Node(Empty, 6, (Node (Empty, 4, Empty))))
```

(a) (b) (c) (d) (e) none of the above

Answer: `(c)`

```
let t2 : int tree =
 insert (insert (insert (insert Empty 4) 5) 6) 7
```

(a) (b) (c) (d) (e) none of the above

Answer: `(a)`

```
let t3 : int tree =
 insert (insert (insert (insert Empty 2) 5) 3) 6
```

(a) (b) (c) (d) (e) none of the above

Answer: none of the above

```
let t4 : int tree =
 Node(Empty, 3, Node(Node (Empty, 2, Empty), 6, (Node (Empty, 7, Empty))))
```

(a) (b) (c) (d) (e) none of the above

Answer: (d)

c. (15 points) Complete the definition of a function `bst_separate` that, when given an integer x, separates a binary search tree into two parts. The first part should contain the values less than x, the second part should contain the values greater than or equal to x.

For example, when given the binary search tree `t`



the result of `bst_separate 5 t` is the pair of binary search trees:



Your solution must take advantage of the binary search tree invariant to avoid traversing the entire tree and should not refer to any of the bst operations such as `insert`, `remove`, and `inorder`.

(Use the next page for your implementation.)

```
let rec bst_separate (x:int) (t : int tree) : int tree * int tree =
 begin match t with
 | Empty -> (Empty, Empty)
 | Node (l, y, r) ->
   if x = y then (l, Node (Empty, y, r))
   else if x < y then
     let (l1, l2) = bst_separate x l in
     (l1, Node (l2, y, r))
   else
     let (r1, r2) = bst_separate x r in
     (Node (l, y, r1), r2)
```

Grading Scheme:

- *no deduction for minor syntax errors*
- *rough allocation of points: 2 for empty case, 3 for equality case, 5 for $<$ case, and 5 for $>$ case.*
- *-3 recursive calls to *both* left and right subtrees (not using invariant)*
- *-2 (each) missing single recursive call to left or right subtree*
- *-2 (each) not reforming correct tree after recursive call, using the node data constructor*
- *-2 for swapping $<$ and $>$ cases*
- *-1 for missing x argument in recursive call*
- *-1 for not saving result of recursive call to local variable (i.e. recomputing* `bst_separate` *twice as much as necessary)*
- *other errors at discretion*